Inversion of analytic functions via canonical polynomials: a matrix approach

Philip Feinsilver*
René Schott †

Abstract.

An alternative to Lagrange inversion for solving analytic systems is our technique of dual vector fields. We implement this approach using matrix multiplication that provides a fast algorithm for computing the coefficients of the inverse function. Examples include calculating the critical points of the sinc function. Maple procedures are included which can be directly translated for doing numerical computations in Java or C. [1]

*Department of Mathematics, Southern Illinois University, Carbondale, IL 62901 USA
†IECN and LORIA, Université Henri Poincaré, 54506 Vandoeuvre-lès-Nancy, France
[1] A preliminary version of this paper has been presented at AISC 2006

1 Introduction

We review our approach to the local inversion of analytic functions. It is important to note that this is different from Lagrange inversion and is based on the flow of a vector field associated to a given function. It appears to be theoretically appealing as well as computationally effective.

Acting on polynomials in x, define the operators

$D = \dfrac{d}{dx}$ and $X =$ multiplication by x.

They satisfy commutation relations $[D, X] = I$, where I, the identity operator, commutes with both D and X. Abstractly, the Heisenberg-Weyl algebra is the associative algebra generated by operators $\{A, B, C\}$ satisfying the commutation relations

$$[A, B] = C\,, \quad [A, C] = [B, C] = 0\,.$$

The *standard* HW algebra is the one generated by the realization $\{D, X, I\}$.

A system of polynomials $\{y_n(x)\}_{n\geq 0}$ is an *Appell system* if it is a basis for a representation of the standard HW algebra with the following properties:

1. y_n is of degree n in x
2. $D\,y_n = n\,y_{n-1}$.

Such sequences of polynomials are "canonical polynomial systems" in the sense that they provide polynomial representations of the Heisenberg-Weyl algebra, in realizations different from the standard one. Our idea [2, 1] is to illustrate explicitly the rôle of vector fields and their duals, using operator calculus methods for working with the latter.

The main observation is

the action of the dual vector field on exponentials is identical to that of the vector field.

Acting iteratively with a vector field on polynomials, the coefficient functions immediately become involved in successive differentiations, while acting iteratively with the dual vector field always produces polynomials from polynomials. So the dual vector field is preferred for calculations.

Specifically, fix a neighborhood of 0 in $\mathbf{C}$. Take an analytic function $V(z)$ defined there, normalized to $V(0) = 0$, $V'(0) = 1$. Denote $W(z) = 1/V'(z)$ and $U(v)$ the inverse function, i.e., $V(U(v)) = v$, $U(V(z)) = z$.

Now, $V(D)$ is defined by power series as an operator on polynomials in x. We have the commutation relations

$$[V(D), X] = V'(D)\,, \qquad [V(D), XW(D)] = I\ .$$

In other words, $V = V(D)$ and $Y = XW(D)$ generate a representation of the HW algebra on polynomials in x. The basis for the representation is $y_n(x) = Y^n 1$. Recursively

$$y_{n+1} = Y y_n\,, \qquad V y_n = n\, y_{n-1} \tag{1}$$

That is, Y is the *raising operator* and V is the corresponding *lowering operator*. Thus, $\{y_n\}_{n\geq 0}$ form a system of *canonical polynomials*. The operator of multiplication by x is given by

$$X = YV'(D) = YU'(V)^{-1}$$

which is a *recursion operator* for the system.

We identify vector fields with first-order partial differential operators. Consider a variable A with corresponding partial differential operator ∂_A. Given V as above, let $\tilde{Y}$ be the vector field $\tilde{Y} = W(A)\,\partial_A$. As any operator function of D on e^{Ax} acts as multiplication by the corresponding function in the variable A, we observe the following identities

$$\tilde{Y}\, e^{Ax} = W(A)x\, e^{Ax} = xW(A)\, e^{Ax} = xW(D)\, e^{Ax} = Y\, e^{Ax}$$

The important property of these equalities is that Y and $\tilde{Y}$ commute, as they involve independent variables. So we may iterate to get

$$\exp(t\tilde{Y})e^{Ax} = \exp(tY)e^{Ax}\ . \tag{2}$$

On the other hand, we can solve for the left-hand side of this equation using the method of characteristics. Namely, if we solve

$$\dot{A} = W(A) \tag{3}$$

with initial condition $A(0) = A$, then for any smooth function f,

$$e^{t\tilde{Y}} f(A) = f(A(t)) \ .$$

Thus, equation (2) yields

$$\exp(tY) e^{Ax} = e^{xA(t)} \ .$$

To solve equation (3), multiply both sides by $V'(A)$ and observe that we get

$$V'(A)\, \dot{A} = \frac{d}{dt} V(A(t)) = 1 \ .$$

Integrating yields

$$V(A(t)) = t + V(A) \qquad \text{or} \qquad A(t) = U(t + V(A)) \ .$$

Or, writing v for t, we have

$$\exp(vY) e^{Ax} = e^{xU(v+V(A))} \ . \tag{4}$$

We can set $A = 0$, using $V(0) = 0$, to get

$$\exp(vY) 1 = e^{xU(v)}$$

on the one hand while

$$e^{vY} 1 = \sum_{n=0}^{\infty} \frac{v^n}{n!} Y^n 1 \sum_{n=0}^{\infty} \frac{v^n}{n!} y_n(x) \ .$$

In summary, we have the expansion of the exponential of the inverse function

$$e^{xU(v)} = \sum_{n=0}^{\infty} \frac{v^n}{n!} y_n(x)$$

or

$$\sum_{m=0}^{\infty} \frac{x^m}{m!} (U(v))^m = \sum_{n=0}^{\infty} \frac{v^n}{n!} y_n(x) \ . \tag{5}$$

This yields an alternative approach to inversion of the function $V(z)$ rather than using Lagrange's formula. We see that the coefficient of $x^m/m!$ yields the expansion of $(U(v))^m$. In particular, $U(v)$ itself is given by the coefficient of x on the right-hand side.

Specifically, we have:

Theorem 1.1 *The coefficient of $x^m/m!$ in $y_n(x)$ is equal to $\tilde{Y}^n A^m|_{A=0}$, which is the coefficient of $v^n/n!$ in the expansion of $U(v)^m$.*

Proof: Expand both sides of equation (2), using v for t, in powers of x and v, set $A = 0$, interpret $Y^n 1$ as $y_n(x)$, and combine with equation (5), reading right to left, thus:

$$\sum_{n=0}^{\infty} \frac{v^n}{n!} \tilde{Y}^n \sum_{m=0}^{\infty} \frac{x^m}{m!} A^m \bigg|_{A=0} = \sum_{n=0}^{\infty} \frac{v^n}{n!} Y^n 1 = \sum_{n=0}^{\infty} \frac{v^n}{n!} y_n(x) = \sum_{m=0}^{\infty} \frac{x^m}{m!} (U(v))^m$$

□

An important feature of our approach is that to get an expansion of U to a given order requires knowledge of the expansion of W to just that order. The reason is that when iterating $xW(D)$, at step n it is acting on a polynomial of degree $n-1$, so all terms of the expansion of $W(D)$ of order n or higher yield zero acting on y_{n-1}. This allows for streamlining the computations.

Remark. 1. For polynomial V, V' will have polynomial entries, and W will be rational in z. Hence the raising operator will be a rational function of D, linear in x. The coefficients of the expansion of W can be computed directly by finite-step recurrences. These hold for polynomial systems as well.

2. Note that to solve $V(z) = v$ for z near z_0, with $V(z_0) = v_0$, apply the method to $V_1(z) = V(z + z_0) - v_0$, so that $V_1(0) = 0$. The inverse is $U_1(v) = U(v + v_0) - z_0$. Thus $U(v) = z_0 + U_1(v - v_0)$.

2 Matrix Algorithm

First we derive a recurrence for the coefficients of the canonical polynomials. This is interpreted in matrix terms and provides a computational formula.

The matrix approach involves only numerical computations and thus is fast whether executed in Maple or implemented in C or Java.

2.1 Recursion formula for the coefficients of the canonical polynomials

We have a sequence of polynomials $y_n(x)$ satisfying the recursive relations of eq. (1). Write

$$W(z) = w_0 + w_1 z + w_2 z^2 + \cdots + w_k z^k + \cdots$$

Then if f is a polynomial, $f(x) = \sum c_k x^k$, with $c_0 = 0$, we have

$$W(D)f(x) = \sum_{j,k} w_j c_k \, \frac{k!}{(k-j)!} \, x^{k-j}$$

Thus the raising operator $Y = XW(D)$ acts as

$$Y f(x) = \sum_{j,k} w_j c_k \, \frac{k!}{(k-j)!} \, x^{k-j+1} = \sum_{k,l} w_{k+1-l} \, c_k \, \frac{k!}{\Gamma(l)} \, x^l$$

having set $j = k + 1 - l$. Now, let

$$y_n(x) = \sum_{k=1}^{n} c_{nk} x^k$$

Then the above equation yields

$$y_{n+1}(x) = \sum_k c_{n+1\,k} \, x^k = \sum_{k,l} w_{k+1-l} \, c_{nk} \, \frac{k!}{\Gamma(l)} \, x^l$$

Hence,

$$c_{n+1\,l} = \frac{1}{\Gamma(l)} \sum_k c_{nk} \, k! \, w_{k+1-l} \tag{6}$$

2.2 Matrix formulation

Fix an order n. Make an n-vector of the coefficients of the polynomial y_m, $m \le n$, setting $\mathbf{c}_m = (c_{m1}, \ldots, c_{mm}, 0, \ldots, 0)$, padding with zeros as needed.

From eq. (6), we form matrices corresponding to the factors on the right-hand side. The index k goes along the rows, l traversing the columns.

Starting with the coefficients of $W(z)$, define the matrix

$$W = \begin{pmatrix} w_1 & w_0 & 0 & \dots & 0 \\ w_2 & w_1 & w_0 & \dots & 0 \\ \vdots & \vdots & \vdots & \ddots & \vdots \\ w_{n-1} & w_{n-2} & w_{n-3} & \dots & w_0 \\ w_n & w_{n-1} & w_{n-2} & \dots & w_1 \end{pmatrix} .$$

Next, define the diagonal matrices

$$P = \begin{pmatrix} 1! & 0 & \dots & 0 \\ 0 & 2! & \dots & 0 \\ \vdots & \vdots & \ddots & \vdots \\ 0 & 0 & \dots & n! \end{pmatrix}, \quad M = \begin{pmatrix} 1 & 0 & \dots & 0 \\ 0 & 2 & \dots & 0 \\ \vdots & \vdots & \ddots & \vdots \\ 0 & 0 & \dots & n \end{pmatrix},$$

$$Q = \begin{pmatrix} 1/\Gamma(1) & 0 & \dots & 0 \\ 0 & 1/\Gamma(2) & \dots & 0 \\ \vdots & \vdots & \ddots & \vdots \\ 0 & 0 & \dots & 1/\Gamma(n) \end{pmatrix}.$$

noting that $M = QP$.

Normalize $w_0 = 1$. Then $y_0(x) = 1$, $y_1(x) = x$. Thus, we write $\mathbf{c}_1$ as an n-vector

$$c_{1j} = \begin{cases} 1, & j = 1 \\ 0, & j > 1 \end{cases}$$

Start with $\mathbf{c}_1 = \mathbf{e}_1 = (1, 0, \dots, 0)$. Then we can formulate eq. (6) for $m > 1$ as

$$\mathbf{c}_m = \mathbf{c}_{m-1}\, PWQ = \mathbf{e}_1\, (PWQ)^{m-1}$$

Observe that successive applications of PWQ can be written as follows

$$PWQPWQ \cdots PWQ = PWMWM \cdots WQ$$

With $\mathbf{e}_1 P = \mathbf{e}_1 = \mathbf{e}_1 M$, we thus have, e.g., $\mathbf{e}_1 PWQPWQ = \mathbf{e}_1 MWMWQ = \mathbf{e}_1 (MW)^2 Q$, etc. Thus,

Proposition 2.1 *The vector of coefficients of $y_m(x)$ is given by*

$$\mathbf{c}_m = \mathbf{e}_1 (MW)^{m-1} Q$$

For order n approximation, the coefficients $w_1, \ldots, w_n$ are the only ones used and the procedure stops with $\mathbf{c}_n$. We have the algorithm:

1. Start with the unit vector $\mathbf{u}_1 = \mathbf{e}_1$ of length n.
2. Iterate $\mathbf{u}_m = \mathbf{u}_{m-1}\, MW$.
3. At each stage, $\mathbf{v}_m = \mathbf{u}_m\, Q$ gives the coefficients of y_m.
4. Stop after $n-1$ iterations, resulting in $\mathbf{v}_n$.
5. Form the matrix with rows $\mathbf{v}_1, \ldots, \mathbf{v}_n$.
6. Premultiplying by a vector of scaled powers of v and postmultiplying by a vector of powers of x yields the exponential $\exp(xU(v))$ to order n.
7. The expansion to order n of $U(v)^k$ is $k!$ times the coefficient of x^k in the result of step 6.

Note that this yields immediately the expansion of $g(U(v))$ to order n for any polynomial $g(z)$ of degree at most n.

2.3 Maple code

The procedure `invbymatW` implements the above algorithm in Maple (see Appendix). Another procedure `invbymatV`, converts the input V to $W = 1/V'$ and calls `invbymatW`.

3 Examples

We illustrate with three classes of examples.

Example 1. Let $V(z) = z - z^2$. Then $W(z) = 1/(1-2z)$ and invoking invbyMatW to order 10 yields (see Worksheet 1) for the coefficient of x

$$U(v) = v + v^2 + 2\,v^3 + 5\,v^4 + 14\,v^5 + 42\,v^6 + 132\,v^7 + 429\,v^8 + 1430\,v^9 + 4862\,v^{10} + \cdots$$

the generating function for the Catalan numbers. For $U(v)^3$, we multiply the coefficient of x^3 by 3! and get

$$U(v)^3 = v^3 + 3\,v^4 + 9\,v^5 + 28\,v^6 + 90\,v^7 + 297\,v^8 + 1001\,v^9 + 3432\,v^{10} + \cdots$$

An important application is to the class of examples where W is given. This is typical in defining special functions by integrals, and in statistics to find the inverse of cumulative distribution functions. Taking 0 as base point, we have

$$V(z) = \int_0^z \frac{1}{W(\zeta)}\, d\zeta$$

Example 2.
a. Expansion of tan. For $W(z) = 1 + z^2$, to order 10 we find

$$\tan(v) = v + \frac{1}{3}\, v^3 + \frac{2}{15}\, v^5 + \frac{17}{315}\, v^7 + \frac{62}{2835}\, v^9 + \cdots$$

For $\tan^2$,

$$\tan^2(v) = v^2 + \frac{2}{3}\, v^4 + \frac{17}{45}\, v^6 + \frac{62}{315}\, v^8 + \frac{1382}{14175}\, v^{10} + \cdots$$

b. Inverse Gaussian. Consider

$$V(z) = \int_0^z e^{-t^2/2}\, dt$$

without the normalization by $\sqrt{2\pi}$. Or, $W(z) = e^{z^2/2}$ and to order 10 we have

$$U(v) = v + \frac{1}{6}\, v^3 + \frac{7}{120}\, v^5 + \frac{127}{5040}\, v^7 + \frac{4369}{362880}\, v^9 + \cdots$$

A similar approach works for any cumulative distribution function with a locally non-vanishing analytic density.

Example 3. Critical points of the sinc function.
An interesting application is locating the critical points of the sinc function

$$\operatorname{sinc} x = \frac{\sin x}{x}$$

with $\operatorname{sinc} 0 = 1$. A quick calculation shows that the nonzero critical points, ζ, satisfy the equation

$$\sin\zeta = \zeta\,\cos\zeta \tag{7}$$

i.e., solutions of the equation $x = \tan x$. We consider the positive side only. A sketch shows that as $x \to \infty$, the intersection points of the graph of $y = x$ with $y = \tan x$, will approach corresponding multiples of $\pi/2$, poles of $\tan z$.

Let $p_n = (2n+1)\pi/2$, $n = 0, 1, \ldots$, and $\zeta_n \approx p_n$ be the associated critical point of sinc. Write $\zeta_n = p_n - z_n$. Then $\sin p_n = (-1)^n$, $\cos p_n = 0$ so that

$$\sin(p_n - z_n) = (p_n - z_n)\,\cos(p_n - z_n)$$

yields $p_n - z_n = 1/\tan z_n$,

$$p_n = \frac{1 + z_n\,\tan z_n}{\tan z_n}$$

Since $z_n \to 0$ as $p_n \to \infty$, we write

$$\frac{1}{p_n} = \frac{\tan z_n}{1 + z_n\,\tan z_n}$$

Thus, consider

$$V(z) = \frac{\tan z}{1 + z\,\tan z}$$

with $V(0) = 0$ and corresponding inverse U. We have $z_n = U(1/p_n)$. I.e.,

$$\zeta_n = p_n - U(1/p_n)$$

Differentiating yields for $W = 1/V'$,

$$W(z) = (1 + z\,\tan z)^2 = 1 + 2\,z^2 + \frac{5}{3}z^4 + \frac{14}{15}z^6 + \frac{17}{35}z^8 + \frac{682}{2835}\,z^{10} + \cdots$$

which can be readily obtained from the results of Example 2.

Running invbyMatW(W,10) we get

$$\begin{aligned}\exp(xU(v)) &= vx + \frac{1}{2}\,v^2x^2 + \frac{1}{6}\,v^3\left(4\,x + x^3\right) + \frac{1}{24}\,v^4\left(16\,x^2 + x^4\right) \\ &+ \frac{1}{120}\,v^5\left(104\,x + 40\,x^3 + x^5\right) + \frac{1}{720}\,v^6\left(784\,x^2 + 80\,x^4 + x^6\right)\end{aligned}$$

$$
\begin{aligned}
&+ \quad \frac{1}{5040}\, v^7 \left(7008\, x + 3304\, x^3 + 140\, x^5 + x^7\right)\\
&+ \quad \frac{1}{40320}\, v^8 \left(79360\, x^2 + 10304\, x^4 + 224\, x^6 + x^8\right)\\
&+ \quad \frac{1}{362880}\, v^9 \left(899712\, x + 479872\, x^3 + 26544\, x^5 + 336\, x^7 + x^9\right)\\
&+ \quad \frac{1}{3628800}\, v^{10} \left(13723776\, x^2 + 2068480\, x^4 + 59808\, x^6 + 480\, x^8 + x^{10}\right) + \cdots
\end{aligned} \tag{8}
$$

and

$$
U(v) = v + \frac{2}{3}\, v^3 + \frac{13}{15}\, v^5 + \frac{146}{105}\, v^7 + \frac{781}{315}\, v^9 + \cdots
$$

which gives the asymptotic formula

$$
\begin{aligned}
\zeta_n \quad = \quad & \frac{(2n+1)\pi}{2} + \frac{2}{(2\,n+1)\,\pi} + \frac{16}{3\,(2\,n+1)^3\,\pi^3} + \frac{416}{15\,(2\,n+1)^5\,\pi^5}\\
& + \frac{18688}{105\,(2\,n+1)^7\,\pi^7} + \frac{399872}{315\,(2\,n+1)^9\,\pi^9} + \cdots
\end{aligned}
$$

For the critical values, we have from eq. (7), $\operatorname{sinc}\zeta_n = \cos\zeta_n$ or

$$
\begin{aligned}
\cos(\zeta_n) \quad &= \quad \cos(p_n - z_n) = (-1)^n \sin z_n \approx (-1)^n (z_n - \frac{z_n^3}{6} + \cdots)\\
&= \quad (-1)^n (U(1/p_n) - \frac{U(1/p_n)^3}{6} + \cdots)
\end{aligned} \tag{9}
$$

Now we use the fact that the expansion of $U(v)^k$ is $k!$ times the coefficient of x^k in eq. (8). Expanding sine to order 9 in eq. (9), and taking the expansion of powers of U from eq. (8), we get (see Worksheet 2)

$$
\operatorname{sinc}\zeta_n \approx \frac{1}{p_n} + \frac{1}{2\,p_n^3} + \frac{13}{24\,p_n^5} + \frac{61}{80\,p_n^7} + \frac{49561}{40320\,p_n^9}
$$

Table I shows the results of using our approximation for the first 10 critical points. We set `Digits:=30` to assure accurate results from Maple. The displayprecision was set to 17, which is sufficient for this comparison.

The first column is the index, the second and third columns are corresponding values of p_n, multiples of π. In each cell of the fourth column, the top entry is our approximation to ζ_n, the middle entry is the result of calling `fsolve` on the function $x - \tan x$ in Maple, and the bottom entry is the difference between the two. Column five is similar for corresponding critical values, $\operatorname{sinc}\zeta_n$, except that the bottom entry is the relative difference ("relative error") of the top and middle values.

3.1 Worksheets

CatalanOut.mw

```
> read "/home/ph/maple/InverseTools.txt";
```

[CoefficientList, CoefficientVector, GcdFreeBasis, GreatestFactorialFactorization, Hurwitz, IsSelfReciprocal, MinimalPolynomial, PDEToPolynomial, PolynomialToPDE, ShiftEquivalent, ShiftlessDecomposition, Shorten, Shorter, Sort, Split, Splits, Translate]

"ExpandsF(function_of_z, to order N)"

"CanPolysV(function_of_z_to_invert, number of polys to generate)"

"CanPolysW(W = 1 over V', number of polys to generate)"

"kron(A,B) returns the Kronecker product of two matrices"

"cmm(X,Y) returns the commutator of two matrices"

"normalize converts multiplication into matrix multiplication"

"invbymatW(W,order) outputs exponential of xU(v) and U(v) to that order in v"

"invbymatV(V,order) outputs exponential of xU(v) and U(v) to that order in v"

```
> n:=10;
```

$$n := 10$$

```
> unassign('z'):
> V0:=z->z-z^2;
> z0:=0;v0:=V0(0);
> V:=V0(z+z0)-v0;taylor(V,z=0,n+1);
```

$$V0 := z \to z - z^2$$

$$z0 := 0$$

$$v0 := 0$$

$$V := z - z^2$$

$$z - z^2$$

```
> INV:=invbymatV(V,10):EXU:=INV[1];U:=INV[2];U3:=3!*coeff(EXU,x^3);
```

$$EXU := v\,x + \frac{1}{2}v^2\,(2\,x + x^2) + \frac{1}{6}v^3\,(12\,x + 6\,x^2 + x^3) + \frac{1}{24}v^4\,(120\,x + 60\,x^2 + 12\,x^3 + x^4) + \frac{1}{120}v^5\,(1680\,x + 840\,x^2 + 180\,x^3 + 20\,x^4 + x^5) + \frac{1}{720}v^6\,(30240\,x + 15120\,x^2 + 3360\,x^3 + 420\,x^4 + 30\,x^5 + x^6) + \frac{1}{5040}v^7\,(665280\,x + 332640\,x^2 + 75600\,x^3 + 10080\,x^4 + 840\,x^5 + 42\,x^6 + x^7) + \frac{1}{40320}v^8\,(17297280\,x + 8648640\,x^2 + 1995840\,x^3 + 277200\,x^4 + 25200\,x^5 + 1512\,x^6 + 56\,x^7 + x^8) + \frac{1}{362880}v^9\,(518918400\,x + 259459200\,x^2 + 60540480\,x^3 + 8648640\,x^4 + 831600\,x^5 + 55440\,x^6 + 2520\,x^7 + 72\,x^8 + x^9) + \frac{1}{3628800}v^{10}\,(17643225600\,x + 8821612800\,x^2 + 2075673600\,x^3 + 302702400\,x^4 + 30270240\,x^5 + 2162160\,x^6 + 110880\,x^7 + 3960\,x^8 + 90\,x^9 + x^{10})$$

$$U := v + v^2 + 2\,v^3 + 5\,v^4 + 14\,v^5 + 42\,v^6 + 132\,v^7 + 429\,v^8 + 1430\,v^9 + 4862\,v^{10}$$

$$U3 := v^3 + 3\,v^4 + 9\,v^5 + 28\,v^6 + 90\,v^7 + 297\,v^8 + 1001\,v^9 + 3432\,v^{10}$$

sincOut1.mw

> EXU:=invbymatW((1+z*tan(z))^2,10)[1];

$$EXU := v\,x + \frac{1}{2}v^2x^2 + \frac{1}{6}v^3(4\,x + x^3) + \frac{1}{24}v^4(16\,x^2 + x^4) + \frac{1}{120}v^5(104\,x + 40\,x^3 + x^5) + \frac{1}{720}v^6(784\,x^2 + 80\,x^4 + x^6) + \frac{1}{5040}v^7(7008\,x + 3304\,x^3 + 140\,x^5 + x^7) + \frac{1}{40320}v^8(79360\,x^2 + 10304\,x^4 + 224\,x^6 + x^8) + \frac{1}{362880}v^9(899712\,x + 479872\,x^3 + 26544\,x^5 + 336\,x^7 + x^9) + \frac{1}{3628800}v^{10}(13723776\,x^2 + 2068480\,x^4 + 59808\,x^6 + 480\,x^8 + x^{10})$$

> sinf:=convert(taylor(sin(z),z=0,11),polynom);

$$sinf := z - \frac{1}{6}z^3 + \frac{1}{120}z^5 - \frac{1}{5040}z^7 + \frac{1}{362880}z^9$$

> cf:=CoefficientList(sinf,z);

$$cf := \left[0,\ 1,\ 0,\ \frac{-1}{6},\ 0,\ \frac{1}{120},\ 0,\ \frac{-1}{5040},\ 0,\ \frac{1}{362880}\right]$$

> SNCCRIT:=add(cf[k+1]*k!*coeff(EXU,x^k),k=1..9);

$$SNCCRIT := v + \frac{1}{2}v^3 + \frac{13}{24}v^5 + \frac{61}{80}v^7 + \frac{49561}{40320}v^9$$

> n:=4:pn:=(2*n+1)*Pi/2:

> "estimated value of sinc at critical point n=4",subs(v=1/pn,(-1)^n*SNCCRIT);"approximate value",evalf(%[2]);

$$\text{"estimated value of sinc at critical point n=4"},\ \frac{2}{9\,\pi} + \frac{4}{729\,\pi^3} + \frac{52}{177147\,\pi^5} + \frac{488}{23914845\,\pi^7} + \frac{198244}{122037454035\,\pi^9}$$

"approximate value", 0.07091345944998001

3.2 Table I

n	p_n	p_n	z_n / ζ_n / Difference	CritVal / CRITVAL / Rel. Diff.
1	$3\pi/2$	4.712…	4.49340966130587088 4.49340945790906418 $2.0339680670759763\ 10^{-7}$	-0.217233628211221657 -0.217233535858463227 $4.2513104067422898\ 10^{-7}$
2	$5\pi/2$	7.853…	7.72525183763185454 7.72525183693770716 $6.9414737108935947\ 10^{-10}$	0.128374553525899137 0.128374553209389319 $2.4655183496759231\ 10^{-9}$
3	$7\pi/2$	10.995…	10.9041216594457662 10.9041216594288998 $1.6866352284048659\ 10^{-11}$	-0.0913252028230576721 -0.0913252028153587987 $8.4301738891623359\ 10^{-11}$
4	$9\pi/2$	14.137…	14.0661939128325293 14.0661939128314735 $1.0557885974744733\ 10^{-12}$	0.0709134594504621526 0.0709134594499800131 $6.7989843130816989\ 10^{-12}$
5	$11\pi/2$	17.278…	17.2207552719308845 17.2207552719307687 $1.1574104321658600\ 10^{-13}$	-0.0579718023461538856 -0.0579718023461010194 $9.1193006263299137\ 10^{-13}$
6	$13\pi/2$	20.420…	20.3713029592875812 20.3713029592875628 $1.8391005862689000\ 10^{-14}$	0.0490296240140741670 0.0490296240140657656 $1.7135325652490864\ 10^{-13}$
7	$15\pi/2$	23.561…	23.5194524986890104 23.5194524986890065 $3.8059406780869000\ 10^{-15}$	-0.0424796169776126470 -0.0424796169776109082 $4.0931811386726112\ 10^{-14}$
8	$17\pi/2$	26.703…	26.6660542588126745 26.6660542588126735 $9.5980478905820000\ 10^{-16}$	0.0374745199939311803 0.0374745199939307418 $1.1701704427987988\ 10^{-14}$
9	$19\pi/2$	29.845…	29.8115987908929591 29.8115987908929588 $2.8222132922950000\ 10^{-16}$	-0.0335251350213987548 -0.0335251350213986259 $3.8462500886809326\ 10^{-15}$
10	$21\pi/2$	32.986…	32.9563890398224768 32.9563890398224767 $9.3819122961900000\ 10^{-17}$	0.0303291711863102861 0.0303291711863102432 $1.4133855236425763\ 10^{-15}$

4 Appendix

4.1 Maple procedures

The worksheet starts with the global declarations:

```
with(LinearAlgebra);
with(PolynomialTools);
```

The "local" declaration for the procedure has been suppressed for clarity.

```
invbymatW:=proc(ff,order)
unassign('z','v','x'):
f:=convert(taylor(ff,z=0,order+1),polynom);
dg:=degree(f,z);
CL:=CoefficientList(f,z);
if(dg<order) then
for k to (order-dg) do
CL:=[op(CL),0]
od;
fi;
dg:=order;
W:=matrix(dg,dg,(i,j)->if(i-j>=-1) then CL[i-j+2] else 0 fi);
M:=diag(seq(k,k=1..dg));
Q:=diag(seq(1/GAMMA(k),k=1..dg));
MW:=multiply(M,W);
UMAT[1]:=Vector(dg,i->if(i=1) then CL[1] else 0 fi);
VMAT[1]:=Vector(dg,i->if(i=1) then CL[1] else 0 fi);
for i from 2 to dg do
UMAT[i]:=multiply(UMAT[i-1],MW);
VMAT[i]:=multiply(UMAT[i],Q);
od:
vv:=vector([seq(v^i/i!,i=1..dg)]);
xx:=vector([seq(x^k,k=1..dg)]);
EXU:=multiply(vv,multiply(stackmatrix(seq(VMAT[k],k=1..dg)),xx));
EXU, coeff(EXU,x)
end:
```

```
"invbymatW(W,order) outputs exponential of xU(v) and U(v) to that
order in v";
```

and you can input V directly as well:

```
invbymatV:=proc(ff,order) local WW;
unassign('z'):
invbymatW(1/diff(ff,z),order)
end:
"invbymatV(V,order) outputs exponential of xU(v) and U(v) to that
order in v";
```

The code “InverseTools.txt” is available at
http://chanoir.math.siu.edu/MATH/InverseTools .

References

[1] P. Feinsilver and R. Schott. *Algebraic structures and operator calculus, Vols I–III.* Kluwer Academic Publishers, 1993, 1994, 1996.

[2] P. Feinsilver and R. Schott. Vector fields and their duals. *Adv. in Math.*, 149:182–192, 2000.

[3] P. Feinsilver and R. Schott. Operator calculus approach to solving analytic systems. *Artificial Intelligence and Symbolic Computation*, Proceedings 8th International Conference, AISC 2006, Beijing, China, LNAI 4120:170–180, 2006.